
'-------------------------------------------------------------- 

'                (c) 2006 Gerhard Günzel 
'-------------------------------------------------------------- 
'  Datei: alarmanlage.BAS 
'-------------------------------------------------------------- 
'  Beschreibung: Alarmanlage 
' 
'  Kern ist ein Mega168-Board mit I2C Bus zur Relaisplatine. 
'  Das Relaisboard hat 4 Ausgänge und 4 Eingänge, die über einen I2C-Expander 
'  am Mega168-Board hängen. Die 4 Eingäng lösen Int0 aus. Das Mega168-Board 
'  liest dann die Eingänge und schaltet dann die Ausgänge und das Handy-M20. 
'  Eine LCD-Anzeige 24x2 ist als Anzeige für Zustände im Klartext da. 
'  Ein zusätzlicher Transponder löst Int1 aus, der scharf oder nicht scharf 
'  und Reset der Alarme auslöst.Int1 wird auch über ein Relais auf der Relais- 
'  platine erzeugt.Zweimalige Auslösung von Int1 (Transponder) innerhalb 15 
'  Sekunden schaltet scharf oder nicht scharf. Einfache Auslösung von Int1 
'  schaltet wenn Alarm ausgelöst ist alle Alarme aus. 
'  Über Serielle ist ein M20 Terminal an das Mega168-Board angeschlossen. 
'  Im Alarmfall oder bei Netzausfall wird eine SMS geschickt. 
'  Alle wichtigen Zustände wie Eingänge, Ausgänge, Netzausfall, Anlage scharf 
'  werden im EEPROM gespeichert und bei Start der Anlage berücksichtigt. 
'  In EEprom-Adresse 1 ist der Status der Softbits 
'  Bei Stomausfall läuft die Anlage mit Batterie weiter und meldet dies per Handy. 

'  Nach Strom wieder da wird die Anlage wieder scharf gestellt, wenn sie vorher 
'  scharf war. Wenn Strom wieder da ist wird ebenfalls eine SMS gesendet. 
'  Bei Ausserbetriebnahme unbedingt vorher unscharf schalten. 
' 
' 
'   ----- Bit 0 bis 7 sind Zustand des PCF8574-A 
' 
'   BITFOLGE     PCF8574                76543210 
'   Ein-Ausgänge PCF8574                EEEEAAAA 
' 
'   bit 0   A     Sirene aussen         Klemme X1 
'   bit 1   A     Sirene innen          Klemme X2 
'   bit 2   A     Strahler              Klemme X3 
'   bit 3   A     Blinkdiode            Klemmen X8,9,10 
'   bit 4   E     Lichtschranke         Klemme X7 
'   bit 5   E     Bewegungsmelder 2     Klemme X6 
'   bit 6   E     Bewegungsmelder 1     Klemme X5 
'   bit 7   E     Netzrelais            Klemme X4 
' 
'   ----- Bit 0 bis 7 der EEProm-Softwarebits - Zustand Wahr ist 1 
' 

'   BITFOLGE                             76543210 
'   bit 0         Sirene aussen Rel 1           0 
'   bit 1         Sirene innen  Rel 2          1 
'   bit 2         Strahler      Rel 3         2 
'   bit 3         Scharf Blinkd Rel 4        3 
'   bit 4         Alarm                     4 
'   bit 5 
'   bit 6 
'   bit 7         Netzausfall            7 
' 
' 
'------ Standard-Init von BASCOM ------------------------------ 
$regfile = "M168def.dat" 
$baud = 19200 
$crystal = 8000000 
Config Lcdpin = Pin , Db4 = Portb.0 , Db5 = Portc.1 , Db6 = Portc.2 , Db7 = Portc.3 , E = Portd.7 , 
Rs = Portd.6 
Config Lcd = 24 * 2 
Config Scl = Portc.5 
Config Sda = Portc.4 
 

'------------------------------------------------------------- 
Const 8574a_write = &B01110000          'pcf8574  Adresse schreiben 
Const 8574a_read = &B01110001           'pcf8574  Adresse lesen 
Dim 8574a_inputs As Byte                'Variable für PCF8574 
Dim Softbits As Byte                    'Merker Soft-Zustände 8 BITS 
Dim Sekundenzaehler As Byte             'Timer1 Zähler 
Dim Transponderzaehler As Byte 
Dim Durchlaufzaehler_timer0 As Word     'Zähler Timer0 
Transponderzaehler = 0 
 
 
 
' ----- Externe Interrupts ----------------------------------- 
Config Portd.2 = Input                  'Int0 als Eingang 
Portd.2 = 1                             'Pullup ein 
Config Portd.3 = Input                  'Int1 als Eingang 
Portd.3 = 1                             'pullup ein 



 

'------- Unbenutzte Pins als Eingänge und Pullup ------------- 
Config Portc.0 = Input 
Portc.0 = 1 
Config Portd.4 = Input 
Portd.4 = 1 
Config Portd.5 = Input 
Portd.5 = 1 
Config Portb.1 = Input 
Portb.1 = 1 
Config Portb.2 = Input 
Portb.2 = 1 
 
'----- Timer1 als Zeitbasis ----- 
On Timer1 Timer1_einesekunde 
Config Timer1 = Timer , Prescale = 1024 
Enable Timer1 
Stop Timer1 
 
'----- Timer2 als Zeitbasis ----- 
On Timer0 Timer0_zwanzig_minuten 
Config Timer0 = Timer , Prescale = 1024 
Enable Timer0 
Stop Timer0 
 
 
'----- Interrup Routinen festlegen --------------------------- 
On Int0 Ir0_routine 
On Int1 Ir1_routine 
Cls 
'----- Ab hier Start der Alarmanlage 
 
'----- Alle Ralaises aus ------------------------------------- 
I2csend 8574a_write , &B11110000        'Relaises aus 
 
'----- Warteschleife für Bewegungsmelder --------------------- 
'      und wichtig auch für Modem - Init dauert ein wenig 
Locate 1 , 1 
Lcd "Anlage Start - 15 sec." 
Wait 15 
 
'----- M20 Modem Pin eingeben - Modem ins Netz --------------- 
Print "AT+CPIN=1234" 
Wait 3 
'----- M20 Modem Text Modus ---------------------------------- 
Print "AT+CMGF=1" 
Cls 
 
'----- EEprom auslesen ------- 
Readeeprom Softbits , 1                 'Softwarezustand lesen 
 
 
'----- Netzausfall prüfen und bei Ausfall 5 Minuten warten---- 
If Softbits.7 = 1 Then                  'Netzausfallbit prüfen 
   Locate 1 , 1 
   Lcd "Warte 5 Minuten" 
   Locate 2 , 1 
   Lcd "Ob Netz stabil ist" 
   Wait 300                             '5 Minuten 
   Cls 
   Softbits.7 = 0                       'Netzausfallbit rücksetzen 
   Writeeeprom Softbits , 1 
   Print "AT+CMGS={034}+49000000000{034}"       'Meldung als SMS Nullen sind Handynummer 
   Waitms 500 
   Print "Strom ist wieder da{026}" 
   Wait 30 
   Print "ATD+4900000000000"            'Anruf zu Hause Nullen sind Telefonnummer 
   Wait 30 
End If 
 
 
 
'----- Zustand der Ports lesen 
I2creceive 8574a_read , 8574a_inputs    'Port lesen und auch INT 8574 löschen 
Locate 1 , 1 
Lcd "Prt  " : Lcd Bin(8574a_inputs)     'ports anzeigen 
Locate 2 , 1 
Lcd "EEpr " : Lcd Bin(softbits)         'EEprom Software anzeigen 
 
'----- Interrupts generell ein (Statusregister bit 7) 
'      Flanke ist falling Edge 
Set Sreg.7                              'Interrupt global ein.I in Statregister 



Eicra = &B00001010                      'falling edge 

'----- Int1 für Transponder ein Melder aus ----- 
Eimsk = &B00000010                      'Int1(Transp=bit1) ein 
 
'----- Prüfen, ob Anlage vor Netzausfall scharf war ---------- 
If Softbits.3 = 1 Then                  'Anlage war bei Netzausfall scharf 
   8574a_inputs.3 = 1                   'Blinkled an 
   I2csend 8574a_write , 8574a_inputs   'Relais Blinkdiode an 
   Softbits.4 = 0                       'Alarmbit rücksetzen sonst kain Alarm 
   Writeeeprom Softbits , 1 
   Eimsk.0 = 1                          'Int0 an - Scharf stellen 
End If 
 
 
'----- Hauptschleife prüft ob Netz wieder da ist ------------- 
Do 
'I2creceive 8574a_read , 8574a_inputs    'Ports auslesen 
If 8574a_inputs.7 = 1 Then              'Netzrelais angezogen, Netz ist da 
     If Softbits.7 = 1 Then             'Netzausfallbit gesetzt 
      Softbits.7 = 0                    'Netzausfallbit rücksetzen 
      Writeeeprom Softbits , 1 
      Print "AT+CMGS={034}+49000000000{034}"       'Meldung als SMS Nullen sind Handynummer 
      Waitms 500 
      Print "Strom ist wieder da{026}" 
      Wait 30 
      Print "ATD+4900000000000"         'Anruf zu Hause Nullen sind Telefonnummer 
      Wait 30 
     End If 
End If 
Loop 
 
End                                     'end program 
 
 
'----- Interruptroutinen ----- 
'----- Interrupt 0 Meldereingänge ---------------------------- 
Ir0_routine: 
Eimsk.0 = 0                             'Int0 aus für Melder 
Readeeprom Softbits , 1                 'EEprom Lesen 
Locate 2 , 1 
Lcd "EEpr " : Lcd Bin(softbits)         'EEprom Software anzeigen 
I2creceive 8574a_read , 8574a_inputs    'Port lesen und auch INT 8574 löschen 
Locate 1 , 6 
Lcd Bin(8574a_inputs)                   'ports anzeigen 
 
If Softbits.4 = 0 Then                  'Alarm nicht gesetzt 
  If 8574a_inputs.4 = 0 Then            'Lichtschranke 
   Softbits.4 = 1                       'Alarmbit seten 
   Writeeeprom Softbits , 1             'Ins EEprom 
   Durchlaufzaehler_timer0 = 0 
   Print "AT+CMGS={034}+49000000000{034}"       'Meldung als SMS Nullen sind Handynummer 
   Waitms 500 
   Print "Alarm an Lichtschranke{026}" 
   Wait 30 
   Print "ATD+4900000000000"            'Anruf zu Hause Nullen sind Telefonnummer 
   Wait 30 
   I2csend 8574a_write , &B11111111     'Alarm ein alle Relais 
   Start Timer0                         'Alarm für 15 Min. ein 
  End If 
End If 
 
If Softbits.4 = 0 Then                  'Alarm nicht gesetzt 
 If 8574a_inputs.5 = 0 Then             'Bewegungsmelder2 
   Softbits.4 = 1                       'Alarmbit seten 
   Writeeeprom Softbits , 1             'Ins EEprom 
   Durchlaufzaehler_timer0 = 0 
   Print "AT+CMGS={034}+49000000000{034}"       'Meldung als SMS Nullen sind Handynummer 
   Waitms 500 
   Print "Alarm Bewegungsmelder oben{026}" 
   Wait 30 
   Print "ATD+4900000000000"            'Anruf zu Hause Nullen sind Telefonnummer 
   Wait 30 
   I2csend 8574a_write , &B11111111     'Alarm ein alle Relais 
   Start Timer0                         'Alarm für 15 Min. ein 
  End If 
End If 
 
If Softbits.4 = 0 Then                  'Alarm nicht gesetzt 
 If 8574a_inputs.6 = 0 Then             'Bewegungsmelder1 
   Softbits.4 = 1                       'Alarmbit seten 
   Writeeeprom Softbits , 1             'Ins EEprom 
   Durchlaufzaehler_timer0 = 0 



   Print "AT+CMGS={034}+49000000000{034}"       'Meldung als SMS Nullen sind Handynummer 
   Waitms 500 
   Print "Alarm Bewegungsmelder unten{026}" 
   Wait 30 
   Print "ATD+4900000000000"            'Anruf zu Hause Nullen sind Telefonnummer 
   Wait 30 
   I2csend 8574a_write , &B11111111     'Alarm ein alle Relais 
   Start Timer0                         'Alarm für 15 Min. ein 
  End If 
End If 
 
If 8574a_inputs.7 = 0 Then              'Netzausfallrelais 
   Softbits.7 = 1                       'Netzausfallbit setzen 
   Writeeeprom Softbits , 1             'Ins EEprom 
   Print "AT+CMGS={034}+49000000000{034}"       'Meldung als SMS Nullen sind Handynummer 
   Waitms 500 
   Print "Kein Strom in Griechenland{026}" 
   Wait 30 
   Print "ATD+4900000000000"            'Anruf zu Hause Nullen sind Telefonnummer 
  Wait 30 
 
' !call 0 
End If 
 
 
  Eimsk.0 = 1                           'Int0  ein 
  Locate 2 , 1 
  Lcd "EEpr " : Lcd Bin(softbits)       'EEprom Software anzeigen 
Return 
 
'----- Interrupt 1 - Transponder ----------------------------- 
Ir1_routine: 
Eimsk.1 = 0                             'Int1(Transp=bit1)aus 
Waitms 25                               'Relaisprellen 
Eifr.1 = 1                              'Interrupt Flag  aus 
Transponderzaehler = Transponderzaehler + 1 
Timer1 = 57723 
Start Timer1 
Eimsk.1 = 1                             'Int 1 wieder einschalten 
Return 
 
 
'----- Zeitbasis eine Sekunde timer auf 57723 
Timer1_einesekunde: 

 Timer1 = 57723 
 Sekundenzaehler = Sekundenzaehler + 1 
 Gosub 10_sekunden 
Return 
 
10_sekunden: 
 Locate 1 , 18 
 Lcd Sekundenzaehler ; " " ; Transponderzaehler 
  If Sekundenzaehler = 10 Then          'Nach 10 sec prüfen 
     Stop Timer1 
     If Transponderzaehler => 2 Then    'Transponder öfter als einmal 
       Toggle 8574a_inputs.3            'Blinkrelais umschalten 
          If 8574a_inputs.3 = 1 Then    'Wenn Blinkrelais an (scharf) 
             Eimsk.0 = 1                'Int0 an 
             Softbits.3 = 1             'Scharfbit setzen 
             Softbits.4 = 0             'Alarmbit rückseten 
             Writeeeprom Softbits , 1   'Zustand scharf in EEprom Bit 3 
             Locate 2 , 17 
             Lcd "scharf  " 
          Elseif 8574a_inputs.3 = 0 Then       'nicht scharf 
             Eimsk.0 = 0                'Int0 aus 
             8574a_inputs.0 = 0         'Rel1 aus 
             8574a_inputs.1 = 0         'Rel2 aus 
             8574a_inputs.2 = 0         'Rel3 aus 
             Softbits.3 = 0             'Scharfbit rücksetzen 
             Softbits.4 = 0             'Alarmbit rücksetzen 
               Writeeeprom Softbits , 1 'Ins EEprom 
             Locate 2 , 17 
             Lcd "unscharf" 
          End If 
       I2csend 8574a_write , 8574a_inputs 
            Locate 1 , 6 
            Lcd Bin(8574a_inputs)       'ports anzeigen 
     End If 
     If Transponderzaehler = 1 Then     'Transponder einmal = Alle Relais aus 
           8574a_inputs.0 = 0           'Rel1 aus 
           8574a_inputs.1 = 0           'Rel2 aus 
           8574a_inputs.2 = 0           'Rel3 aus 



'           8574a_inputs.3 = 1           'Rel4 ein 

         I2csend 8574a_write , 8574a_inputs 
         Locate 1 , 6 
         Lcd Bin(8574a_inputs)          'ports anzeigen 
         Stop Timer0                    'Bei Reset vom Alarm per Transponder 
         Softbits.4 = 0                 'Alarmbit rücksetzen 
         Writeeeprom Softbits , 1       'Ins EEprom 
         Eimsk.0 = 1                    'Int0  wieder ein 
     End If 
    Transponderzaehler = 0 
    Sekundenzaehler = 0 
  End If 
  Locate 2 , 1 
  Lcd "EEpr " : Lcd Bin(softbits)       'EEprom Software anzeigen 
Return 
 
'----- Langzeitschleife 15 Minuten --------------------------- 
Timer0_zwanzig_minuten: 
'Print Durchlaufzaehler_timer0           'Testanzeige 
  Durchlaufzaehler_timer0 = Durchlaufzaehler_timer0 + 1 
  If Durchlaufzaehler_timer0 = 2746 Then       '27465 durchläufe = 15 Min. 
     Stop Timer0 
     Softbits.4 = 0                     'Alarmbit aus 

     Writeeeprom Softbits , 1           'Ins EEprom 
     8574a_inputs.0 = 0                 'Rel1 aus 
     8574a_inputs.1 = 0                 'Rel2 aus 
     8574a_inputs.2 = 0                 'Rel3 aus 
     I2csend 8574a_write , 8574a_inputs 
     Eimsk.0 = 1                        'Int0  wieder ein 
     Durchlaufzaehler_timer0 = 0 
  End If 
Return 


